UNIVERSIDADE FEDERAL DO RIO GRANDE DO SUL

INSTITUTO DE MATEMÁTICA
DEPARTAMENTO DE MATEMÁTICA PURA E APLICADA
Disciplina: MAT01168 -Matemática Aplicada -Semestre Letivo 2008/2
Professoras: Elisabeta Gallicchio e Irene Strauch
TERCEIRA ÁREA

INTEGRAL DE FOURIER UTILIZANDO O MAPLE

Nona aula 19/11/2008

Integral de Fourier Trigonométrica

Seja f(t) definida para todo t, seccionalmente contínua em todo intervalo finito e tal que $\int_{-\infty}^{\infty} |f(t)|\,dt < \infty$, então a função tem sua reprentação integral Fourier Trigonométrica como

$$f(t) := \frac{1}{\pi}\left(\int_0^{\infty} [A(w)\cos(w\,t) + B(w)\sin(w\,t)]\,dw\right),$$

onde os coeficientes integrais são dados por

$$A(w) = \int_{-\infty}^{\infty} f(t)\cos(w\,t)\,dt \quad \text{e} \quad B(w) = \int_{-\infty}^{\infty} f(t)\sin(w\,t)\,dt$$

Exemplo 1: (a) Escreva a função $f(t) := \begin{cases} 0 & t < -1 \text{ and } -t < -1 \\ 1 & -t < 1 \text{ and } t < 1 \end{cases}$ ou $f(t) := \begin{cases} 1 & |t| \le 1 \\ 0 & \textit{otherwise} \end{cases}$ em termos da representação integral de Fourier trigonométrica. (b) Determine os valores da integral obtida. (c) Com o resultado, obtenha a integral de Dirichlet $\int_0^{\infty} \left[\frac{1\,\sin(w)}{w}\right] dw = \frac{\pi}{2}$.

Para o item (a): A função dada é aperiódica e definida para $-\infty < t < \infty$

```
> restart:with(plots):with(inttrans):
Warning, the name changecoords has been redefined
```

```
> f(t):=piecewise(t<-1 and t>1,0,t>-1 and t<1,1);
```

$$f(t) := \begin{cases} 0 & t < -1 \text{ and } -t < -1 \\ 1 & -t < 1 \text{ and } t < 1 \end{cases} \qquad \textbf{(1.1)}$$

Ou

```
> f(t) := piecewise(abs(t) <=1, 1);
```

$$f(t) := \begin{cases} 1 & |t| \le 1 \\ 0 & \textit{otherwise} \end{cases} \qquad \textbf{(1.2)}$$

Pontos de descontinuidade da função:

```
> pont_des:=discont(f(t),t);
```

$$pont\_des := \{-1, 1\} \quad (1.3)$$

Gráfico da função

```
> plot(f(t),t=-2..2,thickness=2,scaling=constrained,titlefont=
  [COURIER,DEFAULT,14],labels=["t", "f(t)"],labelfont=[COURIER,
  DEFAULT,16],axesfont=[COURIER,DEFAULT,16],title=`f(t)=1 se |t|<=
  1,0 para outros valores de t`);
```



Teste da convergência da integral:

```
> teste_conv:=int(abs(f(t)),t = -infinity .. infinity);
```

$$teste\_conv := 2 \quad (1.4)$$

Cálculo dos coeficientes A(w) e B(w)

```
> A(w):= int(f(t)*cos(w*t),t = -infinity .. infinity);
```

$$A(w) := \frac{2\sin(w)}{w} \quad (1.5)$$

```
> B(w):=int(f(t)*sin(w*t),t = -infinity .. infinity);
```

$$B(w) := 0 \quad (1.6)$$

```
> repres:=(1/Pi)*Int([A(w)*cos(w*t)+B(w)*sin(w*t)],w=0..infinity);
```

$$repres := \frac{1}{\pi}\left(\int_0^{\infty}\left[\frac{2\sin(w)\cos(w\,t)}{w}\right]\mathrm{d}w\right) \quad (1.7)$$

(b) Cálculo da média dos valores dos limites laterais da função nos pontos de descontinuidade, isto é, em x= -1 e x=1

```
> media1:=(limit(f(t),t=-1,right)+limit(f(t),t=-1,left))/2;
```

$$media1 := \frac{1}{2} \quad (1.8)$$

```
> media2:=(limit(f(t),t=1,right)+limit(f(t),t=1,left))/2;
```

$$media2 := \frac{1}{2} \quad (1.9)$$

Portanto, a resposta do item (b) é

```
> repres=piecewise(t=-1 , media1,t=1,media2, abs(t)<1,1,0);
```

$$\frac{1}{\pi}\left(\int_0^\infty \left[\frac{2\sin(w)\cos(w\,t)}{w}\right]\mathrm{d}w\right)=\begin{cases}\frac{1}{2} & t=-1\\ \frac{1}{2} & t=1\\ 1 & |t|<1\\ 0 & otherwise\end{cases} \quad \textbf{(1.10)}$$

(c) A **integral de Dirichlet** é obtida, substituindo-se t=0 (e, para simplificar, multiplicamos por Pi e no integrando, vamos forçar a divisão por 2) na última integral

```
> integral_Dirichlet:=expand(Pi)*(eval(subs(t=0,2=1,repres)))=Pi/2;
```

$$integral\_Dirichlet := \int_0^\infty \left[\frac{\sin(w)}{w}\right]\mathrm{d}w=\frac{\pi}{2} \quad \textbf{(1.11)}$$

Esta última integral é o limite da conhecida como integral seno, ou seja, é o $\lim_{x\to\infty}\ \mathrm{Si}(x)=\int_0^x \frac{\sin(t)}{t}\,\mathrm{d}t$

$(-\infty < t < \infty)$

A notação no Maple é `Si(x)=int(sin(t)/t, t=0..x)`. Devido a sua frqüente ocorrência nas aplicações, os valores desta função estão tabulados. Para saber mais, digite >?inifunction

==

Exemplo 2: Mostre que $\frac{1}{\pi}\left(\int_0^\infty \left[-\frac{2\cos\left(\frac{\pi\,w}{2}\right)\cos(w\,t)}{w^2-1}\right]\mathrm{d}w\right)=\begin{cases}\cos(t) & |t|<\frac{\pi}{2}\\ 0 & \frac{\pi}{2}<|t|\end{cases}$. **(1.12)**

Solução:

Define-se f(t) como sendo igual ao lado direito da identidade acima, ou seja,

$$f(t)=\begin{cases}\cos(t) & |t|<\frac{\pi}{2}\\ 0 & \frac{\pi}{2}<|t|\end{cases}.$$

A função é par e se anula fora do intervalo [-π/2 , π/2].

```
> restart:with(plots):with(inttrans):
Warning, the name changecoords has been redefined
```

```
> f(t):=piecewise(abs(t)<Pi/2, cos(t), abs(t)>Pi/2,0);
```

(1.13)

$$f(t) := \begin{cases} \cos(t) & |t| < \frac{\pi}{2} \\ 0 & \frac{\pi}{2} < |t| \end{cases} \tag{1.13}$$

Gráfico da função

```
> plot(f(t),t=-Pi..Pi,thickness=2,scaling=constrained,titlefont=
  [COURIER,DEFAULT,14],labels=["t", "f(t)"],labelfont=[COURIER,
  DEFAULT,16],axesfont=[COURIER,DEFAULT,16],title=`f(t)=cos(t) se
  |t|<Pi/2,0 se |t|>Pi/2`);
```



Para o teste da convergência da integral, deve-se observar que esta função se anula para |x|>π/2. Assim, a integral se reduz ao intervalo -π/2 a π/2.

```
> teste_conv:=int(abs(f(t)),t = -Pi/2 .. Pi/2);
```

$$teste\_conv := 2 \tag{1.14}$$

Cálculo dos coeficientes A(w) e B(w). Como a função é par, sabe-se " a priori" que B(w)=0 e que

$A(w) = \int_{-\infty}^{\infty} f(t)\cos(w\,t)\,\mathrm{d}t = 2\left(\int_{0}^{\infty} f(t)\cos(w\,t)\,\mathrm{d}t\right)$

Deste modo, tem-se:

Só para testar:

```
> B(w):=int(f(t)*sin(w*t),t = -infinity .. infinity);
```

$$B(w) := 0 \tag{1.15}$$

```
> A1(w):= int(f(t)*cos(w*t),t = -infinity .. infinity);
```

$$A1(w) := -\frac{2\cos\left(\frac{\pi w}{2}\right)}{w^2 - 1} \tag{1.16}$$

ou

```
> A(w):= 2*int(f(t)*cos(w*t),t =0 .. infinity);
```

$$A(w) := -\frac{2\cos\left(\frac{\pi w}{2}\right)}{w^2 - 1} \tag{1.17}$$

A(w) não está definida em w=1. Então A(1) deve ser calculado separadamente

```
> A(1):= 2*int(f(t)*cos(1*t),t =0 .. infinity);
```

$$A(1) := \frac{\pi}{2} \tag{1.18}$$

Agora, cabe perguntar se A(w) é contínua em w=1. Em outras palavras, averiguar se **(1.19)**

$$\lim_{w \to 1} \left(- \frac{2 \cos\left(\frac{1 \pi w}{2}\right)}{w^2 - 1} \right) = \frac{\pi}{2}$$

```
> limite:=limit(A(w),w=1);
```

$$limite := \frac{\pi}{2} \tag{1.20}$$

Assim, a representação integral Fourier trigonométrica da função é

```
> repres:=(1/Pi)*Int([A(w)*cos(w*t)+B(w)*sin(w*t)],w=0..infinity);
```

$$repres := \frac{1}{\pi} \left(\int_0^{\infty} \left[- \frac{2 \cos\left(\frac{\pi w}{2}\right) \cos(w\, t)}{w^2 - 1} \right] \mathrm{d}w \right) \tag{1.21}$$

Portanto, substituido-se a representação pela f(t), decorre a identidade

```
> repres=f(t);
```

$$\frac{1}{\pi} \left(\int_0^{\infty} \left[- \frac{2 \cos\left(\frac{\pi w}{2}\right) \cos(w\, t)}{w^2 - 1} \right] \mathrm{d}w \right) = \begin{cases} \cos(t) & |t| < \frac{\pi}{2} \\ 0 & \frac{\pi}{2} < |t| \end{cases} \tag{1.22}$$

$A(w) := - \frac{2 \cos\left(\frac{1 \pi w}{2}\right)}{w^2 - 1}$, A(0)=2, A(1)=$\frac{\pi}{2}$ e decresce a zero para w->∞ . Esta é uma função oscilatória

```
> limit(A(w),w=infinity);
```

$$0 \tag{1.23}$$

Agora, vamos averiguar os zeros de A(w), examinando o numerador e sabendo que, devido ao denominador, w deve se ser diferente de 1, isto é, os pontos onde o gráfico de A(w) intercepta o eixo w

```
> with(Student[Calculus1]):

> Roots( cos(Pi*w/2),2..15 );
```

$$[3, 5, 7, 9, 11, 13, 15] \tag{1.24}$$

Observe atentamente o gráfico abaixo!

```
> plot(A(w),w=0..5*Pi,thickness=2,titlefont=[COURIER,DEFAULT,14],
  labels=["x", "f(x)"],xtickmarks=[3,5,7,9,11,13,15],labelfont=
  [COURIER,DEFAULT,16],axesfont=[COURIER,DEFAULT,16],title=`A(w)=
  -2*cos((Pi/2)*w)/(w^2-1)`);
```



==

As integrais Fourier seno e co-seno

Seja f definida em $[0, \infty)$, seccionalmente contínua e tal que $\int_0^{\infty} |f(t)|\, dt$ converge. A **integral Fourier co-seno** da função é definida como $\frac{1}{\pi}\int_0^{\infty} A(w)\cos(w\,t)\, dw$ e a **integral Fourier seno** como $\frac{1}{\pi}\int_0^{\infty} A(w)\sin(w\,t)\, dw$, onde $A(w) = 2\left(\int_0^{\infty} f(t)\cos(w\,t)\, dt\right)$ e $B(w) = 2\left(\int_0^{\infty} f(t)\sin(w\,t)\, dt\right)$.

A integrais co-seno e seno convergem para f(t) nos pontos onde f é contínua e para $\frac{1}{2}$ (f(t+)+f(t-)) em cada ponto t>0 onde a função apresenta descontinuidade. Em t=0, a integral co-seno converge para f(0+), se f tem derivada à direita em t=0 e a integral seno para 0.

Exemplo: Seja $f(t) = e^{-k\,t}$, $0 \le t$ e k>0. Obtenha a representação integral Fourier co-seno e seno da função.

```
> restart:with(plots):with(inttrans):
Warning, the name changecoords has been redefined
```

```
> assume(k>0):assume(t>=0);interface(showassumed=0);
> f(t):=exp(-k*t);
```

$$f(t\sim) := e^{-k\sim t\sim} \tag{2.1}$$

Teste de convergência da integral

```
> teste_conv:=int(abs(f(t)),t = 0 .. infinity);
```

$$teste\_conv := \frac{1}{k\sim} \tag{2.2}$$

```
> A(w):=radsimp(2*int(f(t)*cos(w*t),t = 0 .. infinity));
```

$$A(w) := \frac{2\,k\sim}{k\sim^2 + w^2} \tag{2.3}$$

A função $f(t) = e^{-kt}$ é diferenciável, então contínua para t>=0 , e a derivada à direita de t=0 existe, assim, a sua representação integral Fourier co-seno para t maior e igual a 0 é

```
> f(t)=(1/Pi)*expand(Int(A(w)*cos(w*t),w = 0 .. infinity));
```

$$e^{-k\sim t\sim} = \frac{2\,k\sim}{\pi}\left(\int_0^\infty \frac{\cos(w\,t\sim)}{k\sim^2 + w^2}\,dw\right) \tag{2.4}$$

Também, pode-se obter o valor da integral abaixo:

```
> expand((1/(2*k))*Int(A(w)*cos(w*t),w = 0 .. infinity))=(Pi/(2*k))
  *f(t);
```

$$\int_0^\infty \frac{\cos(w\,t\sim)}{k\sim^2 + w^2}\,dw = \frac{1}{2}\,\frac{\pi\,e^{-k\sim t\sim}}{k\sim} \tag{2.5}$$

Uma outra maneira é calcular A(w) a integral com o comado direto do Maple. Só que neste caso, aparece o coeficiente $\frac{\sqrt{2}}{\sqrt{\pi}}$ na frente da integral. Veja

```
> convert(fouriercos(f(t),t,w),int);
```

$$\frac{\sqrt{2}\,k\sim}{\sqrt{\pi}\,\left(k\sim^2 + w^2\right)} \tag{2.6}$$

Então, para eliminar este problema devemos colocar $\sqrt{\pi}\,\sqrt{2}$ antes do comando. Ou seja,

```
> A(w):=convert(sqrt(Pi)*sqrt(2)*fouriercos(f(t),t,w),int);
```

$$A(w) := \frac{2\,k\sim}{k\sim^2 + w^2} \tag{2.7}$$

Para a representação integral seno

```
> B(w):=radsimp(2*int(f(t)*sin(w*t),t = 0 .. infinity));
```

$$B(w) := \frac{2\,w}{k\sim^2 + w^2} \tag{2.8}$$

A função tem como sua representação integral Fourier seno

```
> f(t)=(1/Pi)*expand(Int(B(w)*sin(w*t),w = 0 .. infinity));
```

$$e^{-k\sim t\sim} = \frac{2}{\pi}\left(\int_0^\infty \frac{w\sin(w\,t\sim)}{k\sim^2 + w^2}\,dw\right) \tag{2.9}$$

Também, pode-se obter o valor da integral para t>0 abaixo:

```
> expand((1/2)*Int(B(w)*sin(w*t),w = 0 .. infinity))=(Pi/2)*f(t);
```

$$\int_0^\infty \frac{w\sin(w\,t\sim)}{k\sim^2 + w^2}\,dw = \frac{1}{2}\,\pi\,e^{-k\sim t\sim} \tag{2.10}$$

Uma outra maneira é calcular B(w) a integral com o comado direto do Maple. Só que neste caso, aparece o coeficiente $\frac{\sqrt{2}}{\sqrt{\pi}}$ na frente da integral. Veja

```
> convert(fouriersin(f(t),t,w),int);
```

$$\frac{\sqrt{2}\,w}{\sqrt{\pi}\,(k\sim^2 + w^2)} \tag{2.11}$$

Então, para eliminar este problema devemos colocar $\sqrt{\pi}\,\sqrt{2}$ antes do comando. Ou seja,

```
> B(w):=convert(sqrt(Pi)*sqrt(2)*fouriersin(f(t),t,w),int);
```

$$B(w) := \frac{2\,w}{k\sim^2 + w^2} \tag{2.12}$$

Observação: Para t=0 ambas as integrais, acima, não são iguais a f(0)=exp(0)=1, mas convergem para e^{-kt} para todo t>0

Fenômeno de Gibbs e Representação Integral de Fourier

Em analogia à soma parcial dos termos S_n das series de Fourier, define-se a integral Fourier parcial como

$$S_\nu(t) = \frac{1}{\pi}\left(\int_0^\nu [A(w)\cos(w\,t) + B(w)\sin(w\,t)]\,dw\right)$$

E, de acordo com as condições de Dirichlet, tem-se $\lim_{\nu\to\infty} S_\nu(t) = \frac{1}{2}$ (f(t+)+f(t-))

Para exemplificar a convergência, vamos voltar ao exemplo 1, onde

$f(t) := \begin{cases} 0 & t < -1 \text{ and } -t < -1 \\ 1 & -t < 1 \text{ and } t < 1 \end{cases}$, $A(w) := -\frac{2\cos\left(\frac{\pi w}{2}\right)}{w^2 - 1}$ e a representação integral Fourier

da função

$$f(t) = \frac{1}{\pi}\left(\int_0^{\infty}\left[\frac{2\sin(w)\cos(w\,t)}{w}\right]\mathrm{d}w\right)$$

```
> restart: with(inttrans):with(plots):with(plottools):
Warning, the name changecoords has been redefined

Warning, the name arrow has been redefined
```

```
> f(t):=piecewise(t<-1 and t>1,0,t>-1 and t<1,1);
```

$$f(t) := \begin{cases} 0 & t < -1 \text{ and } -t < -1 \\ 1 & -t < 1 \text{ and } t < 1 \end{cases} \tag{3.1}$$

```
> g1:=plot(f(t),t=-2..2,thickness=2,titlefont=[COURIER,DEFAULT,14],
  labels=["t", "f(t)"],labelfont=[COURIER,DEFAULT,14],axesfont=
  [COURIER,DEFAULT,10],title=`f(t)=1 se |t|<=1,0 para outros
  valores de t`):g1;
```



Definindo a integral soma parcial

```
> S[nu](t):=(2/Pi)*Int(((1/w)*sin(w)*cos(w*t)),w = 0 .. nu);
```

$$S_{\nu}(t) := \frac{2}{\pi}\left(\int_0^{\nu}\frac{\sin(w)\cos(w\,t)}{w}\,\mathrm{d}w\right) \tag{3.2}$$

Observe o comando combine expressão trig

```
> combine(2*sin(x)*cos(y),trig);
```

$$\sin(x+y) + \sin(x-y) \tag{3.3}$$

Vamos usar este comando na integral acima

```
> S[nu](t):=combine(expand(S[nu](t),trig));nops(%);
```

$$S_{\nu}(t) := \int_{0}^{\nu} \frac{\sin(w + w\,t) - \sin(-w + w\,t)}{\pi\,w}\,\mathrm{d}w \tag{3.4}$$

$$2$$

Substituindo (u=w*(1+t) e u=-w*(1-t) na integral, obtém-se

```
> S[nu](t):=(1/Pi)*((Int(sin(u)/u,u=0..nu*(1+t)) + Int(sin(u)/u,u=
  0..nu*(1-t))));
```

$$S_{\nu}(t) := \frac{\int_{0}^{\nu(t+1)} \frac{\sin(u)}{u}\,\mathrm{d}u + \int_{0}^{\nu(1-t)} \frac{\sin(u)}{u}\,\mathrm{d}u}{\pi} \tag{3.5}$$

A seguir vamos calcular a soma parcial das integrais, mais precisamente $S_{\nu}(t)$ para υ =1,2,3,4,5,6

```
> for i from 1 to 5 do S[i](t):=evalf(1/Pi)*(Si(i*(1+t))+ Si(i*(1-
  t))):od;
```

$$S_{1}(t) := 0.3183098861\,\mathrm{Si}(t+1) - 0.3183098861\,\mathrm{Si}(-1+t) \tag{3.6}$$

$$S_{2}(t) := 0.3183098861\,\mathrm{Si}(2\,t+2) - 0.3183098861\,\mathrm{Si}(-2+2\,t)$$

$$S_{3}(t) := 0.3183098861\,\mathrm{Si}(3\,t+3) - 0.3183098861\,\mathrm{Si}(-3+3\,t)$$

$$S_{4}(t) := 0.3183098861\,\mathrm{Si}(4\,t+4) - 0.3183098861\,\mathrm{Si}(-4+4\,t)$$

$$S_{5}(t) := 0.3183098861\,\mathrm{Si}(5\,t+5) - 0.3183098861\,\mathrm{Si}(-5+5\,t)$$

```
> Graf:=plot({S[1](t),S[2](t),S[3](t),S[4](t),S[5](t)},t=-2*Pi..2*
  Pi,thickness=2,titlefont=[COURIER,DEFAULT,10],labels=["t","S[N]
  (t)"],labelfont=[COURIER,DEFAULT,16],axesfont=[COURIER,DEFAULT,
  10]): display({Graf,g1},labels=["t","S[N](t)"],xtickmarks=[-6,-5,
  -4,-3,-2,-1,1,2,3,4,5,6],title=`Aproximação por Fourier Integrais
  e o Fenômeno de Gibbs`);
```


Aproxima��o por Fourier Integrais e o Fen�mer
S[N](t)
t

==
==============================